
GP

G.PHILIPPE

Résolution de l'équation de Laplace

par la méthode des différences finies

( méthode à 4 points )

Introduction

On résout l'équation de Laplace: LAPLACIEN (f(x,y))=0 par une méthode numérique. Le problème sera 
invariant selon z. La fonction f est alors dite "harmonique" en mathématiques.

On va discrétiser le problème pour appliquer la méthode des différences finies ( les dérivées sont 
approximées à partir de différences ). 

La méthode est une méthode itérative ( boucle de calcul ) de relaxation ( conditions à vérifier par la fonction 
rendues plus faciles à gérer par la méthode ).

On choisit un problème d'électrostatique. On veut déterminer le potentiel dans une région vide. Le volume 
étudié est limité par quatre plans conducteurs portés à des potentiels connus: 
en x=0 le potentiel vaut VB ( B comme bas )
en x=L le potentiel vaut VH ( H comme haut )
en y=0 le potentiel vaut VG ( G comme gauche )
en y=L le potentiel vaut VD ( D comme droite ).

Rappels théoriques

En électrostatique dans le vide: LAPLACIEN (V) = 0.

Expression de LAPLACIEN (V(x,y)) en fonction des dérivées de V:

LAPLACIEN (V(x,y))= ∂
∂ 2

x2 V   +  ∂
∂ 2

y2 V

Formule approchée ( h désigne le pas de calcul ) utilisée :
V(x,y) = 1/4 ( V(x+h,y) + V(x-h,y) + V(x,y+h) + V(x,y-h)
Pour obtenir cette formule, on écrit le développement de Taylor à l'ordre 2 de V(x+h,y) et de V(x-h,y) 

et on en déduit l'expression de ∂
∂ 2

x2 V .

On fait le même travail pour déterminer l'expression de ∂
∂ 2

y2 V .

On obtient la formule proposée.

Le programme MAPLE
Introduction
> restart; with(plots): Digits:=5:
Warning, the name changecoords has been redefined

On discrétise en faisant un maillage du domaine étudié : 
on se donne un réseau de N*N nœuds régulièrement répartis dont on va déterminer le potentiel. 
Chaque nœud est repéré par ses coordonnées. 
On note h la distance entre deux nœuds successifs selon Ox ou selon Oy.

i=1 correspond à x=0 (bas)
i=N correspond à x=L (haut)
j=1 correspond à y=0 (gauche)
j=N correspond à y=L (droite)

On remarquera que x ne vaudra pas i*h
et que y ne vaudra pas j*h.

Aspect informatique : on stocke les valeurs aux différents nœuds sous forme de tableau ( array ). 
En fait, on utilise deux tableaux : V et old_V qui contient les valeurs du tour précédent. 
L'instruction pour copier un tableau dans un autre est: copy . 
L'instruction pour visualiser les valeurs des éléments du tableau est: eval . 

Initialisation
Déclarations de constantes

On déclare les valeurs numériques des potentiels VG, VD, VB, VH. 
Tout est pris par exemple à zéro, sauf VG choisi égal à 100.00.
Les quatre coins du domaine sont au potentiel zéro.

> VG:=100.00;VD:=0;VB:=0;VH:=0;

 := VG 100.00

 := VD 0

 := VB 0

 := VH 0

Le côté L du domaine carré sera pris égal à 100.
La valeur de N sera prise à 33 
( mais il peut être plus prudent de tester d'abord le programme en utilisant un nombre bien plus petit 
pour voir rapidement si il tourne correctement, par exemple N=9)

> L:=100;N:=33;
 := L 100

 := N 33

Expression littérale du pas h en fonction de N et L:

> h:=evalf(L/(N-1));
 := h 3.1250

On représentera par la suite des vecteurs de norme constante donnant la direction du champ électrique. 
On choisit une norme inférieure à h pour être sûr que deux vecteurs voisins ne se recouvrent pas.

> norme:=0.8*h;
 := norme 2.5000

On décide de déterminer les potentiels à ( environ )  0.01 V près
(Au départ, pour tester le programme, il est prudent de choisir une imprécision plus grande).

> erreurmax:=0.01;
 := erreurmax .01

Il ne doit plus apparaitre désormais aucune valeur numérique dans la suite du programme
puisque toutes les constantes ont été déclarées au départ du programme.

Déclarations de types

On doit déclarer les tableaux pour que Maple connaisse la dimension souhaitée.
Cette déclaration se fait en utilisant  array  .
Pour les potentiels aux noeuds:

> V:=array(1.. N,1..N) ; 
 := V ( )array , , .. 1 33  .. 1 33 [ ]

> old_V:= array(1.. N,1..N) ;                   

 := old_V ( )array , , .. 1 33  .. 1 33 [ ]

Le champ sera calculé mais uniquement aux points intérieurs au domaine
( pas sur la frontière ).
Les résultats seront stockés dans un tableau pour la composante scalaire E_x 
et dans un autre tableau pour la composante scalaire E_y.
Les dimensions des tableaux sont donc différentes.

> E_x:=  array(2.. N-1,2..N-1) ;
 := E_x ( )array , , .. 2 32  .. 2 32 [ ]

> E_y:= array(2.. N-1,2..N-1) ;
 := E_y ( )array , , .. 2 32  .. 2 32 [ ]

Initialisation du tableau de potentiels V
Fixation des valeurs à la frontière

On attribue pour les points appartenant à la frontière les valeurs fixées
qui ne devront pas être modifiées par le programme.

Puisque l'on connait le nombre depoints, il s'agit de boucle  for .
for... from... to... do... end do.
 

Rappel: 
Un élément de tableau se note V[i,j]

> for  i from 1 to N do
         V[i,1]:=VG;
         V[i,N]:=VD;
end do:
for j from 1 to N do
         V[1,j]:=VB;
         V[N,j]:=VH;
end do:

Initialisation des valeurs à l'intérieur du domaine

On attribue pour les autres points de l'intérieur du domaine une valeur initiale égale à zéro.
Le programme va modifier ces valeurs pour obtenir finalement une valeur approchée du potentiel 
aux points du domaine intérieur.

> for  i from 2 to N-1 do
      for j from 2 to N-1 do
                V[i,j]:=0;
      end do:
end do:
Visualisation du tableau V pour vérification

Si on a choisi un N assez petit pour tester, il est intéressant de visualiser le tableau.
On utilise eval . 

> # eval(V);

Calculs
Boucle de calcul du potentiel

La méthode utilisée nous indique que le potentiel en un point est 
la moyenne des potentiels aux quatre nœuds les plus proches:
 V[i,j]=1/4*(V[i+1,j]+V[i-1,j]+V[i,j+1]+V[i,j-1])
Pour chaque nœud intérieur, on va alors calculer une nouvelle valeur de potentiel 
en utilisant les valeurs actuelles pour les quatre nœuds voisins. 

Après un tour complet ( on a balayé tout le tableau V ) , 
on regarde si l'écart le plus grand entre ancienne et nouvelle valeur pour les éléments du tableau 
est inférieur ou supérieur à l'écart max tolérable décidé au départ.
Si l'écart est supérieur, on recommence la boucle.
Si l'écart est inférieur, le calcul du potentiel est supposé achevé.

Quelle instruction MAPLE faut-il utiliser 
pour écrire la boucle?

La structure de base d'un boucle en MAPLE est:
do <instructions> od
Ceci génère en fait une boucle infinie
d'ou l'ajout du for ou du while qui vont limiter cette boucle

1) Lorsque le nombre de tours de boucle est connu à l'avance, on utilise un :
pour i=1 à N faire <instructions> fin du pour
Ce n'est pas le cas ici.

2) Lorsque le nombre de tours à effectuer est inconnu, on a deux possibilités

2a) si il y a un test en entrée de boucle, on utilise alors
tant que <condition d'entrée> faire <instructions> fin du tant que
En MAPLE, c'est le while...do...end do
Ce n'est pas le cas ici. Pas de test en entrée mais en sortie de boucle

2b) si il y a un test en sortie de boucle, on utilise alors
répéter <instructions> jusqu'à ce que <condition de sortie> fin du répéter
En MAPLE, cette structure n'est pas vraiment prévue.
On peut programmer une boucle infinie ( do ... end do ) dans laquelle 
on place, en fin des instructions, la condition de sortie de boucle, en utilisant break . 
do...if... then break end if end do

Quelle instruction MAPLE faut-il utiliser 
pour déterminer l'écart maximum au cours d'un tour complet?

On pourrait faire à chaque fois un test entre l'ancien écart max et le nouvel écart 
qu'on vient de déterminer pour le V[i,j] actuel.
Mais le coût des tests est coûteux 

Mieux vaut donc éviter une instruction telle que: 
si valeur absolue de (V[i,j] - old_V[i,j] ) est supérieur à écart (ancienne valeur), 
alors écart ( nouvelle valeur) = valeur absolue de (V[i,j] - old_V[i,j] ). 

Mieux vaut utiliser: 
ecart = maximum de ( ecart, valeur absolue de (V[i,j] - old_V[i,j] ) )

Voir  max et abs .

Il faut au départ d'un tour de boucle 
1) recopier V dans old_V
2) remettre l'écart à zéro

> do 
     old_V := copy(V):
     ecart:=0:     
     for i from 2 to N-1 do 
          for j from 2 to N-1 do 
               V[i,j]:=1/4*(V[i+1,j]+V[i-1,j]+V[i,j+1]+V[i,j-1]):
               ecart := max(ecart,abs(V[i,j]-old_V[i,j])) :
          end do:
     end do: 
     if ecart < erreurmax then break end if:
od:

Si on a choisi un N assez petit pour tester, il peut être intéressant 
de visualiser old_V et le résultat V .

> #eval(old_V);
> #eval(V);

Calcul du champ

On calcule le champ pour les points intérieurs au domaine étudié.
On utilise toujours une méthode de différence finie en partant des 4 points les plus proches du point 
considéré

> for i from 2 to (N-1) do 
    for j from 2 to (N-1) do 
        E_x[i,j] := -1/(2*h)*(V[i+1,j]-V[i-1,j]):

        E_y[i,j] := -1/(2*h)*(V[i,j+1]-V[i,j-1]): 
    end do:
end do: 

Représentation graphique
Introduction

Les calculs étant terminés, on va visualiser 
les lignes équipotentielles 
et des vecteurs (de norme constante) donnant la direction et le sens du champ. 

Potentiel

Pour le potentiel, on travaille en fait en 3D : 
on représente la fonction V(x,y) en partant des points calculés 
( V sera selon un axe des z ) 
en fonction de x et de y.

On n'utilise pas ici l' instruction habituelle plot3D pour construire un graphe à partir de fonctions
puisque nous avons déjà calculé les coordonnées des points de la surface V(x,y).  
Il nous faut, par contre, stocker ces points déjà calculés sous la forme d'une structure accessible par 
MAPLE
permettant l'affichage par l'instruction PLOT3D ( en majuscules et non pas en minuscules ).
L'instruction à utiliser pour stocker les points obtenus par maillage de la surface étudiée est   MESH 
( maillage) .
MAPLE pourra tracer la surface s'appuyant sur ces points [x,y,z].

Par exemple:
MESH(
[     [ [0,0,0],[0,1,3],[0,2,5] ],
      [ [1,0,0],[1,1,4],[1,2,7] ],
      [ [2,0,0],[2,1,2],[2,2,3] ],
      [ [3,0,0],[3,1,1],[3,2,0] ]          ]  
             )
donnera la surface suivante:
> PLOT3D(MESH([     [ [0,0,0],[0,1,3],[0,2,5] ],
                                    [ [1,0,0],[1,1,4],[1,2,7] ], 
                                    [ [2,0,0],[2,1,2],[2,2,3] ],  
                                    [ [3,0,0],[3,1,1],[3,2,0] ]    
                              ] ),AXESSTYLE(NORMAL));

Il est interessant de comprendre la surface tracée en la faisant tourner avec la souris.

Ici:

> surface_de_points:=MESH([seq([seq(  [(i-1)*h,(j-1)*h,V[i,j]]   , j=1..N )], i=1..N )]):
> tracé_équipotentielles:=PLOT3D(surface_de_points,AXESSTYLE(NORMAL)):
tracé_équipotentielles;

On veut visualiser les équipotentielles. 
En cliquant sur le dessin précédent avec le bouton droit de la souris, on fait apparaitre le menu contextuel. 

COLOR: ZHUE ( la couleur est fonction de z, c'est à dire de V ). 
STYLE: PATCHCONTOUR ( trace sur la surface les lignes de potentiel constant ). 

On choisit une représentation avec des axes orthonormés, dans projection SCALING: CONSTRAINED 
de telle façon que l'axe des x et celui des y aient même échelle. 

On observe le résultat ( c'est le résultat cherché ) 
en vue de dessous ( en faisant faire tourner à la souris 
jusqu'à l'ORIENTATION theta=0 et phi=180 
pour obtenir l'axe des z vers l'arrière et l'axe des x vertical )

Finalement, en une seule instruction: 

> 
tracé_équipotentielles:=PLOT3D(surface_de_points,AXESSTYLE(NORMAL),STYLE(PATCHCONTOUR),CO
LOR(ZHUE),SCALING(CONSTRAINED),ORIENTATION(0,180)):tracé_équipotentielles;

Champ

Pour tracer les vecteur champ, on utilise une instruction MAPLE toute faite : arrow. Voir arrow .
Par exemple: 
on donne le point de départ du vecteur 
puis les composantes du vecteur permettant de déterminer l'extrémité du vecteur.

> arrow([
              [[2,1,0],[1,1,3]],
              [[0,0,4],[2,2,6]]
            ],
              shape=arrow, color=black, axes=normal);
Faire tourner la figure pour mieux comprendre

Si on ajoute l'information de longueur du vecteur, les vecteurs tracés ont toujours la direction précédente 
mais la longueur des vecteurs est désormais fixée

> arrow([
              [[2,1,0],[1,1,3]],
              [[0,0,4],[2,2,6]]
            ],
              shape=arrow, color=black, axes=normal, length=2);

 
On trace donc les vecteurs champ et on constatera que le résultat est peu pratique si le champ est trop petit. 

Ici:

> vecteurs_champ:=arrow([seq(seq(  [[(i-1)*h,(j-1)*h,0],[E_x[i,j],E_y[i,j],0]], j=2..N-1 ), i=2..N-1 )], shape=arrow, 
color=black, axes=normal,scaling=constrained,orientation=[0,180]):vecteurs_champ;

On trace plutôt des vecteurs de norme constante donnant la direction du champ

> direction_champ:=arrow([seq(seq(  [[(i-1)*h,(j-1)*h,0],[E_x[i,j],E_y[i,j],0]], j=2..N-1 ), i=2..N-1 )], 
shape=arrow, color=black, axes=normal,scaling=constrained,orientation=[0,180], length=norme):direction_champ;

Graphe final Champ+Potentiel

Graphe final : on trace équipotentielles et direction du champ sur le même graphe. On a agrandi le tracé 
avec la souris.

> display([direction_champ,tracé_équipotentielles]);

